Université Paris Diderot — Programmation Fonctionnelle
Informatique — Année 2015-2016

TP n°4 - Correction

Logique propositionnelle en OCAML

On considère les formules de la logique propositionnelle construites à partir des variables propositionnelles et les connecteurs $\neg$, $\wedge$ et $\vee$.

Pour représenter une formule en Caml, on se servira du type suivant :

```
type formule  =   Vrai | Faux
                | Var of string
                | Neg of formule
                | Et of formule * formule
                | Ou of formule * formule;;
```

Par exemple, la formule $Vrai \wedge \neg(p \vee r)$ sera représentée par :

```
let f = Et (Vrai, Neg(Ou(Var "p",Var "r")));;
```

Exercice 1. Écrire une fonction

```
string_of_formule : formule -> string
```

qui, étant donnée une formule, renvoie une chaîne de caractères qui représente cette formule écrite avec parenthèses. Par exemple, pour la formule `f` ci-dessus on obtiendra `"(Vrai Et Neg((p Ou r)))"`.

Correction :

```
let rec string_of_formule f = match f with
  | Vrai -> "Vrai"
  | Faux -> "Faux"
  | Var s -> s
  | Neg f -> "Neg("^string_of_formule f^")"
  | Et(f1,f2) -> "("^string_of_formule f1^" Et "^string_of_formule f2^")"
  | Ou(f1,f2) -> "("^string_of_formule f1^" Ou "^string_of_formule f2^")";;
```

Exercice 2. Écrire une fonction

```
list_of_vars : formule -> string list
```

qui, étant donné une formule, renvoie une liste de toutes ses variables (sans élément dupliqué) dans l'ordre alphabétique. Par exemple, `list_of_vars (Et (Ou(Var "p",Var "q"), Var "q"))` renvoie `["p"; "q"]`

Correction : On utilise `union_sorted` du TP2.

```
let rec union_sorted l1 l2 = match l1,l2 with
    | _,[] -> l1
    | [],_ -> l2
    | a1::l1',a2::l2' ->
        if a1 < a2 then a1::(union_sorted l1' l2 ) else
```

```
        if a2 < a1 then a2::(union_sorted l1  l2') else
        a1::(union_sorted l1' l2');;

let rec list_of_vars f =
    match f with
    | Var s -> [s]
    | Neg f -> list_of_vars f
    | Et(f1,f2) | Ou(f1,f2) -> union_sorted (list_of_vars f1)
                                            (list_of_vars f2)
    | _ -> [];;
```

Exercice 3. Appelons *environnement* toute liste d'association de type `(string * bool) list`. Un environnement permet de spécifier les valeurs des variables d'une expression. Chaque variable peut apparaître une seule fois dans l'environnement. Par exemple, pour la formule `f` ci-dessus un environnement est `[("p",true),("r",false)]`

La *valeur d'une expression* `e` *dans l'environnement* `l` est soit indéfinie, si toutes les variables apparraissant dans `e` ne sont pas définies, sinon la valeur booléenne obtenue en remplaçant dans `e` les variables par les valeurs spécifiées par `l`, ainsi que les constantes par leur valeur booléene et tous les connecteurs logiques par les opérations qui leur sont naturellement associées. Écrire une fonction

```
eval_formule : formule -> (string * bool) list -> bool
```

qui évalue une formule étant donnée un environnement.
Par exemple, `eval_formule f [("r",true);("p",false)]` donnera `false`.

Vous pouvez utiliser `List.assoc v l` qui renvoie la partie droite `f` du premier couple de `l` de la forme `(v, f)` s'il existe, et déclenche une exception si ce couple est introuvable.

Correction :

```
let rec eval_formule f l = match f with
    Vrai -> true
  | Faux -> false
  | Var s -> (List.assoc s l)
  | Neg f1 ->  not (eval_formule f1 l)
  | Et(f1,f2) -> (eval_formule f1 l) && (eval_formule f2 l)
  | Ou(f1,f2) -> (eval_formule f1 l) || (eval_formule f2 l);;
```

Simplification de formule

Une formule peut ne contenir aucune variable – elle est alors appelée *formule constante*. La valeur d'une telle formule est directement calculable, sans environnement. Même lorsque une formule contient des variables, certaines parties de cette formule peuvent être constantes : dans ce cas, la formule peut être simplifiée en remplaçant ces sous-formules par leurs valeurs.

Exercice 4. Écrire une fonction

```
eval_sous_formule : formule -> formule
```

telle que `eval_sous_formule f` renvoie la formule obtenue en remplaçant chaque sous-formule constante de `f` par sa valeur. On notera que si `f` est elle-même une formule constante, la valeur renvoyée doit être la formule `f`. D'autre part, si `e1` et `e2` sont constantes, alors `Neg(e1)`,

`Et(e1,e2)` et `Or(e1,e2)` sont aussi des expressions constantes, et peuvent être remplacées par leur valeur. Par exemple, `eval_sous_formule (Et (Var "p", Ou(Faux,Faux)))` donne `Et (Var "p", Faux)`.

Correction :

```
let rec eval_sous_formule f = match f with
  | Et(f1,f2) -> begin
    match (eval_sous_formule f1, eval_sous_formule f2) with
    | Faux,Faux -> Faux | Faux,Vrai -> Faux
    | Vrai,Faux -> Faux | Vrai,Vrai -> Vrai
    | (f'1,f'2) -> Et(f'1,f'2)
  end
  |  Ou(f1,f2) -> begin
    match (eval_sous_formule f1, eval_sous_formule f2) with
    | Faux,Faux -> Faux | Faux,Vrai -> Vrai
    | Vrai,Faux -> Vrai | Vrai,Vrai -> Vrai
    | (f'1,f'2) -> Ou(f'1,f'2)
  end
  | Neg(f) -> begin
    match eval_sous_formule f with
    | Vrai -> Faux | Faux -> Vrai
    | f' -> Neg f'
  end
  | f -> f;;
```

Exercice 5. Même lorsqu'elles sont non constantes, certaines parties d'une formule peuvent malgré tout être simplifiées au moyen des règles suivantes. Etant donnée une sous-expression d'une expression quelconque, on peut la remplacer :

- par `Vrai`, si elle est de la forme `Ou(Vrai,f)` ou `Ou(f,Vrai)`,
- par `Faux`, si elle est de la forme `Et(Faux,f)` ou `Et(f,Faux)`,

Écrire une fonction

```
simplifie_formule : formule -> formule
```

qui implémente les règles ci-dessous.

Par exemple, `simplifie_formule (Et (Var "p", Ou(Faux,Faux)))` donne `Faux`.
($\star$) Donner d'autres règles de simplification et réaliser-les. Par exemple, $A \wedge A$ est équivalent à A.

Correction :

```
let simplifie_formule f = match f with
  | Et(f1,f2) -> begin
    match (eval_sous_formule f1, eval_sous_formule f2) with
    | Faux,_ -> Faux
    | _,Faux -> Faux
    | Vrai,Vrai -> Vrai
    | (f'1,f'2) -> Et(f'1,f'2)
  end
  |  Ou(f1,f2) -> begin
    match (eval_sous_formule f1, eval_sous_formule f2) with
    | Vrai,_ -> Vrai
    | _,Vrai -> Vrai
```

```
    | Faux,Faux -> Faux
    | (f'1,f'2) -> Ou(f'1,f'2)
  end
  | Neg(f) -> begin
    match eval_sous_formule f with
    | Vrai -> Faux | Faux -> Vrai
    | f' -> Neg f'
  end
  | f -> f;;
```

Forme normale conjonctive

Dans la logique propositionnelle les propriétés suivantes sont vérifiés :

1. *Descente des négations vers les variables - Doubles négations et lois de De Morgan.* Les formules suivantes sont équivalentes :
   — $\neg\neg F$ et F
   — $\neg(F \vee G)$ et $(\neg F \wedge \neg G)$,
   — $\neg(F \wedge G)$ et $(\neg F \vee \neg G)$.
2. *Descente des $\vee$ sous les $\wedge$ par factorisation.* Les formules suivantes sont équivalentes :
   — $(F \vee (G \wedge H))$ et $(F \vee G) \wedge (F \vee H)$,
   — $((F \wedge G) \vee H)$ et $(F \vee H) \wedge (G \vee H)$.

Exercice 6. Écrire une fonction

```
fnc : formule -> formule
```

qui transforme une formule vers une formule équivalente en forme normale conjonctive en utilisant les propriétés ci-dessus.

Par exemple, `fnc (Ou (Neg (Et (Var "p", Var "q")), Et (Var "q", Var "r")))` donne :

```
Et
 (Et (Ou (Ou (Neg (Var "p"), Var "p"), Var "q"),
   Ou (Ou (Neg (Var "p"), Var "q"), Var "q")),
 Et (Ou (Ou (Neg (Var "p"), Var "p"), Var "r"),
  Ou (Ou (Neg (Var "p"), Var "q"), Var "r")))
```

Correction :

```
let rec desc_neg f = match f with
    Neg(Neg g) -> desc_neg g
  | Neg(Ou(g,h)) -> Et(desc_neg(Neg g), desc_neg(Neg f))
  | Neg(Et(g,h)) -> Ou(desc_neg(Neg g), desc_neg(Neg f))
  | Ou(g,h) -> Ou(desc_neg g, desc_neg h)
  | Et(g,h) -> Et(desc_neg g, desc_neg h)
  | f -> f
;;

let rec desc_ou f = match f with
    Et(g,h) -> Et(desc_ou g, desc_ou h)
  | Ou(g,h) ->
    let g' = desc_ou g
```

```
    and h' = desc_ou h in
      (match g',h' with
   _,Et(g,h) -> Et(desc_ou (Ou(g',g)),desc_ou (Ou(g',h)))
 | Et(f,g),_ -> Et(desc_ou (Ou(f,h')),desc_ou (Ou(g,h')))
 | _ -> Ou(g',h'))
  | f -> f
;;

let rec fnc f = desc_ou (desc_neg f);;
```

Les tautologies

Une formule de la logique propositionnelle est appelée une *tautologie*, si elle s'évalue à `true` pour tout environnement qui donne des valeurs booléennes à toutes ses variables.

Exercice 7. Écrire une fonction

`environnements : formule -> (string * bool) list list`

qui étant donnée une formule donne une liste de tous ses environnements (une valeur booléenne pour chaque variable de la formule). Par exemple,
`environnements (formule = Et (Vrai, Neg (Ou (Var "p", Var "r")))` donne

```
[[("r", false); ("p", false)]; [("r", false); ("p", true)];
 [("r", true); ("p", false)]; [("r", true); ("p", true)]]
```

Correction :

```
let rec ajouter_element x l = match l with
  | [] -> []
  | h::t -> (x::h)::(ajouter_element x t);;

(* peut se faire mieux *)
let environnements f =
  let l = list_of_vars f in
  let rec aux l l1 = match l with
      [] -> l1
    | h::t -> aux t ((ajouter_element (h,false) l1) @
                     (ajouter_element (h,true) l1)) in
  (aux l [[]]);;
```

Exercice 8. Écrire une fonction

`tautologie : formule -> bool`

qui, étant donnée une formule, détermine si elle est une tautologie ou pas. Par exemple,
`tautologie (Ou(Var "p",Neg(Var "p")))` donne `true`.

Correction :

```
let tautologie f =
  let l = environnements f in
  List.for_all (eval_formule f) l;;
```

L'opération XOR

Exercice 9. ($\star$) Ajouter l'opération XOR aux formules et refaire tous les exercices.